
Wie Sie mit der Excel-Funktion
SORTIEREN() Daten sortiert anzeigen

Mit der Excel-Funktion SORTIEREN() können Sie eine Liste oder einen
Bereich anhand eines Kriteriums automatisch sortieren und an eine
bestimmte Stelle im Tabellenblatt ausgeben.

Zuletzt geändert am 18.03.2026

Die Parameter und Einstellungen für die Funktion
SORTIEREN()

Die Excel-Funktion SORTIEREN() ist eine dynamische Funktion. Ändert sich die
Ausgangsliste oder das Sortierkriterium, dann wird die sortierte Liste automatisch
aktualisiert.

Die Syntax von SORTIEREN() lautet wie folgt:

SORTIEREN(Matrix; Sortierindex; [Sortierreihenfolge]; [Nach_Spalte])

Dabei gilt:

– Matrix: Entspricht dem Datenbereich der Liste, der nach einem Kriterium
sortiert werden soll. Die Kopfzeile ist nicht Bestandteil dieses Bereiches. Im
folgenden Beispiel (unten) ist dies der Datenbereich A2:C11.

– Sortierindex: Eine Zahl, welche die Zeilen- oder Spaltennummer angibt, nach
der sortiert werden soll. Im Beispiel (unten) soll nach der
Abteilungsbezeichnung sortiert werden. Die Abteilungsbezeichnung ist die
dritte Spalte in der Matrix, daher die Zahl 3.

– Sortierreihenfolge (optional): Legt die Art der Sortierung fest. Der Eintrag 1
bedeutet aufsteigende Sortierung, während –1 eine absteigende Sortierung
ergeben würde. Erfassen Sie hier kein Kriterium, dann wird automatisch
aufsteigend sortiert.

– Nach_Spalte (optional): Wenn Sie die Matrix nach Spalten sortieren wollen,
dann erfassen Sie hier den Wert WAHR. Wollen Sie die Matrix nach Zeilen
sortieren, dann erfassen Sie den Wert FALSCH. Wenn Sie hier keinen Wert
erfassen, dann wird standardmäßig nach Zeilen sortiert (FALSCH). Im Beispiel
sind die Zeilen sortiert; entsprechend wird der Standardeintrag verwendet.

Da Sie nur ein Kriterium zur Sortierreihenfolge eingeben können, können Sie die Daten
mit SORTIEREN() nicht nach einem zweiten Kriterium – im Anschluss – sortieren. Eine
Sortierung „erst nach Abteilung“, dann nach „Nachname“ ist also nicht möglich.

Die Funktion SORTIEREN() an einem Beispiel
erklärt

Schauen wir uns die Funktionsweise von SORTIEREN() an einem Beispiel an. In der
folgenden Abbildung finden Sie links die Liste (A1:C11), die automatisch nach der
Abteilungsbezeichnung rechts daneben aufsteigend sortiert dargestellt werden soll.

Daten zum Sortieren

Erfassen Sie hierzu die folgende Formel in die Zelle F5

=SORTIEREN(A2:C11; 3; 1)

Sie bekommen daraufhin automatisch die Liste sortiert nach der
Abteilungsbezeichnung dargestellt.

Mit SORTIEREN() sortierte Liste von Daten

Hinweis: Da SORTIEREN() eine dynamische Funktion ist, braucht sie genug Platz für
die Anzeige des Ergebnisses. Achten Sie deshalb darauf, dass im Ausgabebereich (hier
der Bereich F2:H11) keine anderen Einträge stehen. Ansonsten erhalten Sie die
Fehlermeldung #ÜBERLAUF!

Ändern Sie einen Eintrag in der ursprünglichen Liste im Bereich A2:C11, dann wird dies
sofort auch im sortierten Bereich F2:H11 angezeigt. Die sortierte Liste ist also immer
aktuell und richtig sortiert.

Dynamische Listen mit Leerzeilen sortieren

So können Sie beispielsweise eine Liste für die Dateneingabe führen, in der Sie neue
Einträge einfach unter die bestehenden schreiben. In der Tabelle mit den sortierten
Daten (oder in einem gesonderten Tabellenblatt) haben Sie dann immer das sortierte
Ergebnis.

Dazu erweitern Sie den Bereich, den Sie sortieren wollen, so, dass Sie neue Daten
unten ergänzen können. Beispiel: Sie wollen später weitere Namen ab Zeile 12 (bis
vermutlich Zeile 20) erfassen und im Bereich F1:H20 sortiert ausgeben. Die Sortier-
Formel dazu lautet:

=SORTIEREN(A2:C20; 3; 1)

In der sortierten Liste wird in den Zeilen 12 bis 20 allerdings der Wert 0 angezeigt. Um
das zu verhindern, fangen Sie diese Ausgabe ab mit folgender Formel:

=WENN(A2:A20=""; ""; SORTIEREN(A2:C20; 3; 1; FALSCH))

Und richtig dynamisch wird es, wenn Sie die zu sortierende Tabelle als dynamische
Tabelle anlegen, wie in der folgenden Abbildung mit dem Tabellennamen „Tabelle1“.
Dann kann diese Tabelle beliebig lang werden. Und mit dieser Formel wird sortiert:

=WENN(Tabelle1=""; ""; SORTIEREN(Tabelle1; 3; 1; FALSCH))

Dynamische Tabelle mit SORTIEREN() sortiert

Beim Sortieren in Excel die Groß- und
Kleinschreibung beachten

Die Sortierfunktion in Excel beachtet standardmäßig die Groß- und
Kleinschreibung nicht. Wenn Sie diese Einstellung ändern, können Sie in
einer langen Liste falsche Schreibweisen einfacher erkennen und
korrigieren.

Zuletzt geändert am 18.03.2026

Wenn Sie Daten in Excel sortieren, dann wird die Groß- und Kleinschreibung
standardmäßig nicht beachtet. Das liefert meistens auch das gewünschte Ergebnis.

Manchmal ist es aber nötig, die Groß- und Kleinschreibung beim Sortieren von
Daten zu berücksichtigen. Zum Beispiel dann, wenn Sie die jeweiligen Schreibweisen
in einer langen Liste beisammen sehen wollen, um sie zu korrigieren.

Wie funktioniert die Sortierung mit Groß- und Kleinschreibung?

Liste mit Groß- und Kleinschreibung in einer Spalte
erfassen

Ausgangspunkt ist eine Liste, in der in einer Spalte die Daten fälschlicherweise mal
groß, mal klein geschrieben sind. Wenn Sie dann beim Sortieren die Groß- und
Kleinschreibung beachten, stehen die jeweiligen Schreibweisen direkt untereinander.

Sie finden die falschen Schreibweisen somit en bloc und können sie korrigieren, ohne
dass Sie mühsam in der Liste alle Zellen durchsuchen müssen.

In der folgenden Abbildung sehen Sie eine exemplarische Namensliste, in der manche
Namen fälschlicherweise in Großbuchstaben erfasst sind.

Exemplarische Namensliste mit Groß- und Kleinschreibung

Liste sortieren und Groß- und Kleinschreibung
beachten

Um die Liste zu sortieren, aktivieren Sie als Erstes eine beliebige Zelle in der Liste.

Eine Zelle (A2) in der Liste aktivieren

Wählen Sie anschließend im Excel-Menüband die Befehlsfolge:

Registerkarte Daten > Befehlsgruppe Sortieren und Filtern > Befehl Sortieren

Befehl Sortieren im Menüband aktivieren

Daraufhin öffnet sich das Dialogfeld Sortieren. Sie bekommen in diesem Dialogfeld
grundsätzlich alle Spalten der Liste und die dazugehörigen Sortierkriterien aufgelistet.

Da das Beispiel nur aus einer Spalte besteht, finden Sie hier nur eine Spalte
dargestellt. Klicken Sie in diesem Dialogfeld jetzt auf die Schaltfläche Optionen, die
sich oben in der ersten Zeile befindet.

Schaltfläche Optionen beim Sortieren aktivieren

Es öffnet sich das Dialogfeld Sortieroptionen. Aktivieren Sie hier das
Kontrollkästchen Groß-/Kleinschreibungbeachten, das standardmäßig deaktiviert
ist.

Kontrollkästchen Groß-/Kleinschreibung beachten

Durch Klick auf die Schaltfläche OK schließen Sie das Dialogfeld Sortieroptionen. Sie
gelangen wieder zum Dialogfeld Sortieren zurück.

Starten Sie jetzt den Sortiervorgang, indem Sie das Dialogfeld Sortieren durch Klick
auf die Schaltfläche OK schließen.

Jetzt wird beim Sortiervorgang die Groß- und Kleinschreibung entsprechend beachtet.
Die kleingeschriebenen Namen werden vor dem entsprechenden großgeschriebenen
Namen einsortiert.

Die Namen in Großbuchstaben stehen jetzt alle untereinander und Sie können diese in
einem Schritt per Copy-and-paste korrigieren.

Sortierte Liste mit Beachtung der Groß-/Kleinschreibung

Listen in Excel nach mehreren Kriterien
automatisch sortieren

Mit der Excel-Funktion SORTIERENNACH() können Sie eine Liste oder
einen Bereich nach mehreren Kriterien automatisch sortieren und an
einer anderen Stelle im Tabellenblatt ausgeben. Änderungen bei den zu
sortierenden Daten werden sofort in der sortierten Liste übernommen.

Zuletzt geändert am 18.03.2026

Neben der Funktion SORTIEREN() gibt es in Excel mittlerweile auch die Funktion 
SORTIERENNACH(). Während Sie mit SORTIEREN() eine Liste nach nur einem
Kriterium sortieren können, können Sie mit SORTIERENNACH() eine Liste nach
mehreren Kriterien automatisch sortieren und an einer bestimmten Stelle im
Tabellenblatt ausgeben.

Bei SORTIERENNACH() handelt es sich um eine dynamische Funktion. Ändert sich die
Ausgangsliste oder ein Sortierkriterium, dann wird die sortierte Liste automatisch
aktualisiert.

Aufbau und Parameter der Excel-Funktion
SORTIERENNACH()

Schauen wir uns die Syntax von SORTIERENNACH() genauer an, um die Funktionsweise
besser zu verstehen. Die Syntax lautet wie folgt:

=SORTIERENNACH(Matrix; Nach_Matrix1; [Sortierreihenfolge1];
[Nach_Matrix2]; [Sortierreihenfolge2]; ...) 

Dabei bedeuten:

– Matrix: Datenbereich der Liste, der nach einem oder mehreren Kriterien
sortiert werden soll. Die Kopfzeile der Liste ist nicht Bestandteil von Matrix. Im
folgenden Beispiel befinden sich die Daten im Bereich A2:C11.

– Nach_Matrix1 :Bereich innerhalb der Ausgangsliste, nach dem die Liste als
Erstes sortiert werden soll. Analog zur Matrix darf hier auch nicht die
Kopfzeile angegeben werden. Im Beispiel soll als Erstes nach Nachname
sortiert werden; entsprechend ist dies der Bereich B2:B11.

– Sortierreihenfolge1 (optional): Angabe der Sortierreihenfolge. 1 bedeutet
die aufsteigende Sortierung, –1 die absteigende Sortierung. Erfassen Sie hier
kein Kriterium, dann wird automatisch aufsteigend sortiert.

– Nach_Matrix2 (optional): Bereich innerhalb der Ausgangsliste, nach dem als
Nächstes sortiert werden soll. Analog zur Matrix darf hier auch nicht die
Kopfzeile mitreferenziert werden. Im Beispiel soll nach dem Nachnamen nach
dem Vornamen sortiert werden. Entsprechend ist dies der Bereich A2:A11.

– Sortierreihenfolge2 (optional): Angabe wie Sortierreihenfolge1.

Wollen Sie die Liste nach weiteren Kriterien sortieren, dann erfassen Sie die
zusätzlichen Kriterien und Sortierreihenfolgen analog als zusätzliche Argumente.

Beispiel für die Funktion SORTIERENNACH()

Schauen wir uns die Funktionsweise von SORTIERENNACH() an einem Beispiel an. In
der folgenden Abbildung finden Sie links die Liste (A1:C11), die automatisch nach
Nachnamen und anschließend nach Vornamen rechts daneben aufsteigend sortiert
dargestellt werden soll.

Beispiel für eine Liste, die sortiert werden soll

Erfassen Sie hierzu die folgende Formel in der Zelle F2:

=SORTIERENNACH(A2:C11; B2:B11; 1; A2:A11; 1)

Sie bekommen daraufhin automatisch die Liste sortiert nach Nachnamen und
anschließend nach Vornamen dargestellt.

Nach mehreren Kriterien sortierte Liste

SORTIERENNACH() ist eine dynamische Funktion. Verändern die Sortierkriterien oder
Einträge in der Ausgangsliste, wird die sortierte Liste im rechten Bereich sofort
aktualisiert.

Excel-Liste der Monatsnamen nach Datum
sortieren

Die Reihenfolge der Monate soll der zeitlichen Reihenfolge entsprechen.
Sind die Monate mit ihren Namen dargestellt, ergibt die einfache,
alphabetische Liste ein unerwünschtes Ergebnis. So bringen Sie die
Monatsliste in die chronologische Reihenfolge.

Zuletzt geändert am 18.03.2026

In Excel kann es vorkommen, dass Sie eine Liste mit Monatsnamen chronologisch
anordnen möchten – von Januar bis Dezember.

Aber: Wenn die Monatsnamen als Text vorliegen, versucht Excel standardmäßig eine
alphabetische Sortierung. Das führt dazu, dass „April“ und „August“ oben stehen
(siehe Abbildung).

Reihenfolge der Monate bei alphabetischer Standard-Sortierung

Wie bringen Sie Excel nun eine benutzerdefinierte Sortierung bei, um die Monate
korrekt in der Reihenfolge Januar bis Dezember zu ordnen?

Benutzerdefinierte Sortierung in Excel

Die benutzerdefinierte Sortierung bietet eine einfache Lösung, um eine Monatsliste
richtig zu sortieren. Excel besitzt bereits eine vordefinierte Liste für die Monatsnamen,
die Sie nutzen können, um Ihre Daten von Januar bis Dezember korrekt anzuordnen.

Markieren Sie dazu eine Zelle in der Liste, die Ihre Monatsnamen enthält. Im Beispiel
markieren Sie die Überschrift Monat und aktivieren anschließend im Menüband die
Befehlsfolge Registerkarte Daten > Befehlsgruppe Sortieren und Filtern > Befehl 
Sortieren.

Menü Benutzerdefiniertes Sortieren in Excel

Daraufhin öffnet sich das Dialogfeld Sortieren.

Hier sehen Sie die Möglichkeit, eine Sortierebene zu definieren. Dadurch, dass Sie die
Überschrift Monat vorher markiert haben, wird die zu sortierende Spalte automatisch
erkannt. Sie können dies daran sehen, dass unter Spalte der Wert Monat ausgewählt
ist.

Bei Sortieren nach ist standardmäßig Zellwerte ausgewählt. Dies bedeutet, dass die
jeweiligen Zellwerte maßgeblich für den Sortiervorgang sind.

Klicken Sie jetzt als Nächstes bei Reihenfolge auf das Dropdown-Symbol am rechten
Rand.

Auswahl der Sortierreihenfolge

Im Dropdown-Menü für die Reihenfolge sehen Sie die Option Benutzerdefinierte
Liste… Klicken Sie darauf, um die integrierten Sortierlisten in Excel aufzurufen.

Benutzerdefinierte Liste… auswählen

Daraufhin öffnet sich das Dialogfeld Benutzerdefinierte Listen.

In diesem Dialogfeld finden Sie eine Auswahl an vordefinierten Listen. Unter anderem
finden Sie auch die Monate des Jahres in der richtigen Reihenfolge von Januar bis
Dezember.

Wählen Sie die Liste mit den Monatsnamen (Januar, Februar, März … Dezember) aus
und bestätigen Sie Ihre Auswahl, indem Sie das Dialogfeld mit einem Klick auf OK
schließen.

Hinweis: In diesem Dialogfeld können Sie auch eigene benutzerdefinierte Listen
definieren. Wählen Sie dazu aus der Auswahl in der linken Box „Benutzerdefinierte
Listen:“ den Eintrag „Neue Liste“, der ganz oben steht.

Auswahl der Benutzerdefinierten Liste für die Sortierung der Monatsnamen

Sie gelangen dann wieder zum Hauptsortierfenster zurück und sehen unter 
Reihenfolge die benutzerdefinierte Liste mit den Monatsnamen dargestellt.

Bestätigen Sie Ihre Einstellungen, indem Sie das Dialogfeld durch einen Klick auf OK
schließen.

Seite 18



Kapitel 301 Sortieren und Filtern in Excel
Excel-Liste der Monatsnamen nach Datum sortieren

Gewünschtes Sortierkriterium wählen

Excel sortiert nun die Monatsnamen in der zeitlich korrekten Reihenfolge von Januar
bis Dezember, ohne auf die alphabetische Sortierung zurückzugreifen.

Liste der Monate in chronologischer Reihenfolge

So können Sie Kundenlisten mit Excel nach
Geburtstagen sortieren

Die Standardsortierung in Excel liefert für Datumswerte nicht immer das
gewünschte Ergebnis. Wenn nur Tag und Monat, aber nicht das Jahr, bei
der Sortierung berücksichtigt werden sollen, müssen Sie einen
Zwischenschritt einfügen. So können Sie beispielsweise
Geburtstagslisten erstellen.

Zuletzt geändert am 18.03.2026

Ein Datum kann in Excel mit dem Format Datum angezeigt und sortiert werden. Der
23.12.2018 steht dann in einer nach aufsteigendem Datum sortierten Liste vor dem
02.03.2020.

Für manche Zwecke will man aber nur nach Tag und Monat sortieren und das Jahr
unberücksichtigt lassen. Der 02.03. soll also vor dem 23.12. angezeigt werden.
Diese spezielle Sortierung braucht man zum Beispiel, wenn man nach 
Geburtstagen im laufenden Jahr sortieren will.

Was auf den ersten Blick wie eine einfache Aufgabe aussieht, ist nur mit einem Kniff
möglich.

Beispiel: Liste nach Geburtsdatum sortieren

Sie wollen Ihre Kundenliste nach Geburtsdatum sortieren, sodass die Liste in der
Reihenfolge der im Jahr auftretenden Geburtstage vorliegt?

Die Kundenliste in der folgenden Abbildung ist nach Nachnamen sortiert. Damit Sie
zum Geburtstag eines Kunden aktiv werden können, soll die Liste nach der Reihenfolge
der Geburtstage im laufenden Jahr sortiert dargestellt werden.

Sie wollen deshalb nach dem Datum in Spalte C so sortieren, dass der zuerst im Jahr
anstehende Geburtstag oben steht.

Kundenliste sortiert nach Nachnamen (Auszug)

Die Standard-Sortierung mit Excel bringt nicht das
gewünschte Ergebnis

Die Standardsortierung nach Geburtsdatum zeigt, worin das Problem der
Aufgabenstellung liegt.

Markieren Sie eine Zelle in der Spalte C der Liste (zum Beispiel C2). Aktivieren Sie im
Menüband die Befehlsfolge: Registerkarte Daten > Befehlsgruppe Sortieren und
Filtern > Befehl Aufsteigend sortieren.

Liste nach Geburtsdatum aufsteigend standardmäßig sortiert

Leider entspricht die sortierte Liste nicht der Aufgabenstellung, da die Liste nicht nach
den anfallenden Geburtstagen im laufenden Jahr sortiert dargestellt wird. Dies liegt
daran, dass die Jahreszahl des Geburtsdatums berücksichtigt wird.

Monat und Tag aus dem Datumsfeld berechnen

Damit die Liste richtig dargestellt wird, muss diese nach Geburtsmonat und
anschließend nach dem jeweiligen Tag im Monat sortiert werden. Um dies zu
erreichen, werden zwei Hilfsspalten eingefügt.

Sie können die Monatszahl aus einem Datum mit der Funktion MONAT() berechnen.
Diese Funktion liefert Ihnen für ein Datum im Januar die Monatszahl 1, für Februar 2
und so weiter.

Als Argument müssen Sie der Funktion lediglich einen Datumswert oder einen
entsprechenden Zellbezug mitgeben. Erfassen Sie daher in der Zelle D2 die folgende
Formel:

=MONAT(C2)

In der Zelle C2 steht der Datumswert, aus welchem der Monatswert abgeleitet werden
soll.

Monatszahl aus Geburtsdatum berechnen

Im nächsten Schritt erstellen Sie rechts daneben eine weitere Hilfsspalte, um die
Tageszahl des jeweiligen Monats zu berechnen.

Analog zur Funktion MONAT() gibt es dafür die Funktion TAG() in Excel. Die Funktion 
TAG() liefert für den 01.01. die Tageszahl 1, für den 15.02. die Tageszahl 15 und so
weiter. Erfassen Sie daher in der Zelle E2 die folgende Formel:

=TAG(C2)

Tageszahl aus Geburtsdatum berechnen

Kopieren Sie die beiden Formeln der Hilfsspalten bis zum Ende Ihrer Geburtstagsliste.

Formeln der Hilfsspalten bis zum Ende der Liste kopieren (Auszug)

Liste in Excel nach Monat und Tag sortieren

Jetzt können Sie die Liste nach den anfallenden Geburtstagen im Jahr sortieren lassen.

Markieren Sie hierzu eine Zelle in der Spalte D und aktivieren Sie im Menüband die
Befehlsfolge Registerkarte Daten > Befehlsgruppe Sortieren und Filtern > Befehl 
Sortieren.

Befehl Sortieren ausführen

Es wird das Dialogfeld Sortieren geöffnet. Dadurch, dass Sie eine Zelle in der Spalte D
vorher markiert haben, wird automatisch bei Spalte die Monatsspalte (Monat) als
Sortierkriterium vorgeschlagen.

Die Einstellungen bei Sortieren nach (Zellwerte) und Reihenfolge (Nach Größe
aufsteigend) werden automatisch richtig gesetzt.

Nach dem Monatswert muss die Liste anschließend nach dem Tageswert sortiert
werden. Drücken Sie daher im Dialogfeld Sortieren auf die Schaltfläche Ebene
hinzufügen, um eine weitere Sortierebene oder Sortierregel hinzuzufügen.

Neue Sortierebene in Excel hinzufügen

Hierdurch wird eine neue Sortierebene hinzugefügt.

Wählen Sie hier bei Dann nach den Eintrag Tag aus. Bei Sortieren nach und 
Reihenfolge werden standardmäßig die Einträge Zellwerte und Nach Größe
(aufsteigend) ausgewählt, welche für diese Aufgabenstellung richtig sind.

Bestätigen Sie Ihre Sortiereinstellungen, indem Sie das Dialogfeld durch einen Klick auf
OK schließen.

Sortierkriterien in Excel definieren

Ihre Liste wird daraufhin nach Monatszahl und anschließend nach Tageszahl sortiert.
Sie erhalten somit die Kundenliste nach den anfallenden Geburtstagen im Jahr sortiert
dargestellt.

Nach Geburtstagen im laufenden Jahr sortierte Kundenliste (Auszug)

Mit der Excel-Funktion FILTER() Daten
selektieren und ausgeben

Mit der Funktion FILTER() können Sie Listeneinträge nach bestimmten
Kriterien filtern und an einer anderen Stelle ausgeben. Sie können so
Datensätze flexibel suchen und anzeigen lassen – auch mit mehreren
Suchkriterien.

Zuletzt geändert am 18.03.2026

Mithilfe der Excel-Funktion FILTER() können Sie einen Bereich von Daten anhand der
von Ihnen definierten Kriterien filtern und die zutreffenden Einträge an eine bestimmte
Stelle im Tabellenblatt ausgeben.

FILTER() ist hierbei eine dynamische Funktion. Das heißt, ändert sich die Liste oder
ein Suchkriterium, dann werden die gefilterten Einträge automatisch entsprechend
aktualisiert.

Die Parameter der Funktion FILTER()

Schauen wir uns zunächst die Syntax von FILTER() genauer an, um die Funktionsweise
von FILTER() besser zu verstehen. Die Syntax lautet wie folgt:

=FILTER(Matrix; Einschließen; [Wenn_leer])

Dabei bedeuten:

– Matrix: Entspricht dem Datenbereich (Liste), der alle Einträge enthält, jedoch
ohne die Kopfzeile. Im folgenden Beispiel also der Zellbereich A2:C11.

– Einschließen: Definiert, welche Spalte der Matrix geprüft und mit welchem
Suchkriterium die Einträge verglichen (gefiltert) werden sollen. C2:C11
entspricht im Beispiel der Spalte in der Liste, in der das Suchkriterium
vorkommen kann.
Durch C2:C11=F2 definieren Sie, dass jeder Eintrag im Bereich C2:C11 mit
dem Suchkriterium in der Zelle F2 verglichen wird. Nur für die Einträge im
Bereich C2:C11, die dem Suchkriterium entsprechen, wird dann der
entsprechende Listeneintrag im Ergebnisbereich ausgegeben.

– Wenn_leer (optional): Mit diesem Argument definieren Sie die Ausgabe, wenn
das Suchkriterium in der Liste nicht gefunden wird. Im Beispiel ist definiert,
dass kein Text ausgegeben wird: "".
Machen Sie hier keine Angabe, dann wird die Fehlermeldung #KALK! als
Ergebnis zurückgegeben, falls kein entsprechender Eintrag in der Liste
vorhanden ist.

Was die Funktion FILTER() bewirkt – ein Beispiel

Schauen wir uns die Funktionsweise von FILTER() an einem Beispiel an.

In der folgenden Abbildung finden Sie im linken Bereich (A1:C11) eine Liste mit
Datensätzen, die nun dynamisch nach bestimmten Kriterien gefiltert werden soll.

Das Suchkriterium befindet sich im Bereich (F1:F2) und die zutreffenden Datensätze
sollen im Ergebnisbereich – beginnend bei Zelle F4 – ausgegeben werden.

Datenbereich (Liste), Suchkriterium und Ergebnisbereich für die Funktion FILTER()

Um nur die Listeneinträge der Mitarbeitenden im Ergebnisbereich darzustellen, die in
der IT arbeiten, erfassen Sie die folgende Formel in der Zelle F5:

=FILTER(A2:C11; C2:C11=F2; "")

Sie bekommen daraufhin automatisch nur die Mitarbeitenden aus der Abteilung „IT“ im
Ergebnisbereich dargestellt.

Ergebnis nach dem Filtern der Liste mit FILTER()

Verändern Sie das Suchkriterium in Zelle F2 in „Revision“, wird das Ergebnis im
Bereich F5:H7 sofort angepasst; Ergebnis wäre dann: Ludwig – Hofmann – Revision.

So können Sie FILTER() einsetzen bei mehreren
Suchkriterien

Im ersten Beispiel haben wir nur ein Suchkriterium für das dynamische Filtern
verwendet. Sie können jedoch auch mehrere Kriterien definieren. In diesem Fall
müssen Sie den Multiplikationsoperator (*) im Argument Einschließen verwenden.

Beispiel: Sie wollen alle Werte in der Liste (A2:C11) zurückgeben, die in der Abteilung
„Technik“ beschäftigt sind UND mit dem Nachnamen „Meier“ heißen. Das zweite
Suchkriterium für den Nachnamen fügen Sie im Beispiel in der Zelle G2 ein. Die
angepasste Formel lautet dann:

=FILTER(A2:C11; (C2:C11=F2)*(B2:B11=G2); "")

Wichtig: Die Formeln für die einzelnen Suchkriterien werden hierbei in 
Klammern () eingeschlossen.

Filter-Funktion bei mehreren Suchkriterien

Filtern mit einem dynamischen Listenbereich

Wenn Sie den Listenbereich A1:C11 als Excel-Tabelle definieren, dann wird die
Filterfunktion immer automatisch auf die gesamte Tabelle angewendet. Sie können
also am Ende der Tabelle weitere Datensätze oder Einträge einfügen, die dann beim
Filtern sofort berücksichtigt werden.

Seite 31



Kapitel 301 Sortieren und Filtern in Excel
Mit der Excel-Funktion FILTER() Daten selektieren und ausgeben

Die Formel für die Filterfunktion erweitert sich automatisch. Aus A2:C11 wird dann zum
Beispiel A2:C14.

Dynamische Liste für das Filtern mit der Tabellen-Funktion von Excel

Hinweis: Die Funktion FILTER() filtert Datensätze, wie Sie es vielleicht mit den
Filtereinstellungen in Tabellen kennen. In der vorigen Abbildung können Sie den
Bereich A2:C14 filtern durch Ihre Auswahl in der Titelzeile A1:C1 und dort die
Dropdown-Funktion zum Filtern.

Der besondere Vorteil der Funktion FILTER() ist, dass Sie damit die Ausgabe an eine
andere Stelle bewirken (im Beispiel ab Zelle F4).

Mit Filtern nur ausgewählte Spalten anzeigen

Wenn Sie Daten aus einer Tabelle oder Liste mit der Filterfunktion auswählen, werden
bei der Ausgabe immer alle Spalten aus dem Bezug Matrix dargestellt, die das
gewünschte Kriterium (Einschließen) erfüllen. Im bisherigen Beispiel werden also im
Ergebnisbereich F bis G immer die drei Spalten der ursprünglichen Tabelle A bis C
angezeigt.

Wenn Sie nur ausgewählte Spalten anzeigen lassen wollen, müssen Sie die Funktion
FILTER() zweifach anwenden.

Zunächst filtern Sie das Kriterium, dann die gewünschten Spalten. Die entsprechende
Funktion für das Beispiel oben lautet für den Fall, dass Sie nur die Spalte mit den
Nachnamen anzeigen lassen wollen:

=FILTER(FILTER(Tabelle2; (Tabelle2[Abteilung]=F2)); {0.1.0})

Funktion FILTER() mit Auswahl der Spalten

Zunächst filtern Sie mit der inneren Filterfunktion die Tabelle nach dem Kriterium
Abteilung = Technik.

Dann geben Sie in der äußeren Filterfunktion mit dem Filter-Parameter {0.1.0} an,
dass nur die zweite der drei Spalten ausgegeben werden soll.

Beachten Sie, dass bei der Eingabe der Punkt (.) als Trennzeichen verwendet wird,
wenn dieser nicht als Dezimaltrennzeichen dient.

Excel-Liste filtern unter Beachtung der
Groß- und Kleinschreibung

So wählen Sie Werte aus einer Liste mit der FILTER-Funktion in Excel aus
– und zwar so, dass die Groß- und Kleinschreibung berücksichtigt wird.

Zuletzt geändert am 18.03.2026

Die FILTER-Funktion in Excel ist sehr praktisch, aber sie unterscheidet standardmäßig
nicht zwischen Groß- und Kleinschreibung. Doch auch dieses Problem lässt sich lösen.

Was macht die FILTER-Funktion standardmäßig?

Die FILTER-Funktion in Excel ermöglicht es, nur die Werte anzuzeigen, die bestimmten
Kriterien entsprechen. Die Syntax lautet:

=FILTER(Matrix; Einschließen; [Wenn_leer])

Sie filtern also eine Liste oder Matrix und geben nur die Zeilen aus, die dem Kriterium
unter Einschließen entsprechen. Hier sind die Parameter von FILTER() im Einzelnen
erklärt.

Die folgende Tabelle dient als Beispiel für das Filtern, wenn nun die Groß- und
Kleinschreibung beachtet werden soll.

Liste zum Filtern der gewünschten Werte

Funktion FILTER() für Groß- und Kleinschreibung
optimieren

Excel unterscheidet bei der Standardnutzung der Funktion FILTER() nicht zwischen
Groß- und Kleinschreibung. Um dies zu umgehen, verwenden Sie die Funktion 
IDENTISCH(), die genau dies tut – sie prüft, ob zwei Textzeichenfolgen exakt
übereinstimmen, inklusive Groß- und Kleinschreibung.

Mit der folgenden Formel filtern Sie unter Beachtung der Groß- und Kleinschreibung:

=FILTER(A2:C12; IDENTISCH(A2:A12; E2))

Die Formelbestandteile haben hierbei die folgende Bedeutung:

– A2:C12: Matrix, die gefiltert werden soll

– A2:A12: Bereich der Maske, welche die Suchkriterien enthält

– E2: Zelle, in der das Suchkriterium steht, nach dem gefiltert werden soll (mit
exakter Übereinstimmung)

Die FILTER-Funktion durchsucht den Bereich A2:C12 und gibt nur die Zeilen zurück, die
bestimmten Kriterien entsprechen.

Das zweite Argument der Funktion enthält die Bedingungen, die erfüllt sein müssen,
damit eine Zeile zurückgegeben wird:

=FILTER(A2:C12; IDENTISCH(A2:A12; E2))

Die IDENTISCH-Funktion prüft, ob der Text in jeder Zelle von A2:A12 exakt,
einschließlich Groß- und Kleinschreibung, mit dem Text in der Zelle E2
übereinstimmt.

IDENTISCH(A2:A12; E2) gibt ein Array (eine Liste von Ergebnissen) mit WAHR oder
FALSCH zurück. Jedes Element des Arrays entspricht einer Zelle in A2:A12.

Wenn der Zellinhalt exakt dem Text in Zelle E2 entspricht (mit Beachtung der
Groß-/Kleinschreibung), liefert die IDENTISCH-Funktion den Wert WAHR. Andernfalls
gibt sie den Wert FALSCH zurück.

Die FILTER-Funktion nutzt das Array von WAHR/FALSCH-Werten, das von IDENTISCH
zurückgegeben wird, als Filterkriterium:

– Zeilen in A2:A12, bei denen IDENTISCH den Wert WAHR ergibt, werden in das
Ergebnis der FILTER-Funktion aufgenommen.

– Zeilen mit dem Wert FALSCH werden durch die FILTER-Funktion
ausgeschlossen.

Die folgende Abbildung zeigt das Ergebnis, wenn die Liste nach der Farbe „ROT“ in der
Großschreibung gesucht wird. Entsprechend steht in Zelle E2 das Suchkriterium ROT.

Wird nach einem Kriterium gesucht, das nicht exakt so in der Liste A2:C12 steht, folgt
die Ausgabe „kein Eintrag“.

Filtern einer Liste mit der Funktion IDENTISCH()

Drei beste oder schlechteste Werte in
Excel automatisch filtern

In diesem Excel-Tipp erfahren Sie, wie Sie mit intelligenten Funktionen
und Formeln die drei besten oder schlechtesten Werte aus einer Liste
automatisch filtern können. Ob für Verkaufszahlen, Leistungsanalysen
oder andere Auswertungen – mit diesen Methoden können Sie Ihre
Daten effizienter analysieren und schneller Entscheidungen treffen.

Zuletzt geändert am 18.03.2026

In der täglichen Arbeit mit Excel stehen wir oft vor der Herausforderung, aus großen
Datenmengen die relevantesten Informationen herauszufiltern.

Besonders wenn es darum geht, aus einer Liste die besten oder schlechtesten
Werte zu ermitteln, kann das manuelle Suchen zeitaufwendig und fehleranfällig
werden. Für diesen Zweck bietet Excel verschiedene Möglichkeiten, diese Aufgabe zu
automatisieren.

Beispiel: Flexible Auswertung von Kennzahlen

In der folgenden Abbildung sehen Sie eine Tabelle mit wichtigen Kennzahlen eines
Konzerns, der in verschiedenen Ländern geschäftliche Aktivitäten verfolgt.

Aus dieser Liste sollen nun immer die drei Einträge mit dem größten prozentualen
Wachstum gegenüber dem Vorjahr automatisch an eine andere Stelle gefiltert werden.

Die gefilterten Werte sollen hierbei absteigend sortiert dargestellt werden. Ändern sich
die Werte in der Liste, dann sollen sich die gefilterten Werte automatisch anpassen.

Tabelle mit Kennzahlen zu Umsatz, Marktanteil, Wachstum für Ländergesellschaften

Die drei größten Werte aus einer Liste filtern und
anzeigen

Diese Aufgabenstellung können Sie mit der folgenden Formel lösen:

=SORTIEREN(FILTER(A2:D11; D2:D11>=KGRÖSSTE(D2:D11; 3)); 4; -1)

Die Formel geht hierbei von den folgenden Annahmen aus:

– A2:D11: Bereich der gesamten Liste, aber ohne die Überschriftenzeile

– D2:D11: Bereich in der Liste, in dem die Suchkriterien stehen

Größte Werte filtern mit SORTIEREN(), FILTERN() und KGRÖSSTE()

Hier erfahren Sie, wie die beiden Funktionen SORTIEREN() und FILTERN()
genau funktionieren und welche Einstellungen Sie dort vornehmen.

Schauen wir uns die Funktionsweise der Formel von innen nach außen an.

Mit der Funktion FILTER(A2:D11; D2:D11>=KGRÖSSTE(D2:D11; 3)) wird zunächst
die Liste nach den drei größten Wachstumswerten durchsucht. Das „Geheimnis“ liegt
hier im zweiten Argument der Filterfunktion, D2:D11>=KGRÖSSTE(D2:D11; 3), in
dem Sie die Suchkriterien definieren.

Mit KGRÖSSTE(D2:D11;3) ermitteln Sie zunächst den drittgrößten Wachstumswert.

Hinweis: Wollen Sie anstelle der drei größten die fünf größten Werte ermitteln, dann
tauschen Sie einfach das zweite Argument in der Funktion KKGRÖSSTE() entsprechend
aus. Für die 3 in KGRÖSSTE(D2:D11;3) setzen Sie die Zahl 5 ein.

Anschließend vergleichen Sie diesen Wert mit jedem Wachstumswert im Bereich
D2:D11; das ist das zweite Argument der Filterfunktion: 
D2:D11>=KGRÖSSTE(D2:D11; 3).

Ist der Wert größer oder gleich dem drittgrößten Wachstumswert, dann wird die
entsprechende Zeile mit der Funktion FILTER() gefiltert.

Die drei gefilterten Werte werden dann an die Funktion SORTIEREN() übergeben,
welche die gefilterten Werte absteigend (-1) nach der Wachstumsrate (4. Spalte)
sortiert.

=SORTIEREN(FILTER(A2:D11; D2:D11>=KGRÖSSTE(D2:D11; 3)); 4; -1)

Die drei kleinsten Werte aus einer Liste filtern und
anzeigen

Wollen Sie sich die drei schlechtesten Werte anzeigen lassen, dann müssen Sie die
Formel nur wie folgt anpassen:

=SORTIEREN(FILTER(A2:D11; D2:D11<=KKLEINSTE(D2:D11; 3)); 4; 1)

Kleinste Werte filtern mit SORTIEREN(), FILTERN() und KKLEINSTE()

Hier wird zunächst mit der Funktion KKLEINSTE() der drittkleinste Wert ermittelt.
Anschließend werden die Listeneinträge gefiltert, deren Wert kleiner oder gleich dem
drittkleinsten Wert ist.

Gefilterte Zeilen mit einer Excel-Formel
zählen

Beim Einsatz von Filtern in Excel-Tabellen passen sich Formeln und ihre
Ergebnisse nicht einfach an die gewählten Filterkriterien an. Wenn Sie
die Funktion TEILERGEBNIS() mit den bekannten Rechenoperationen
kombinieren, erhalten Sie das gewünschte Ergebnis. So gehen Sie vor.

Zuletzt geändert am 18.03.2026

Die Aufgabe: Erst filtern, dann zählen

Wenn Sie in einer Excel-Tabelle mit Filtern arbeiten und Zeilen durch Filterkriterien
ausblenden oder einblenden, dann passen sich einfache Formeln nicht auf die jeweilige
Filterauswahl an.

Beispiel: Sie möchten die Anzahl von gefilterten Zeilen in einer Tabelle mit einer
Formel zählen?

Mit den klassischen Funktionen wie ZÄHLENWENN() oder auch ANZAHL2() kommen
Sie hier nicht weiter, da diese zwar alle Zellen in einem bestimmten Bereich zählen,
aber nicht berücksichtigen, wenn Zeilen über einen Filter ausgeblendet werden.

Das Ergebnis würde daher alle Zeilen enthalten (sichtbare und ausgeblendete Zeilen).

Angenommen, Sie haben eine Tabelle mit Umsatzdaten, die Spalten für Monat, Jahr
und Umsatz beinhaltet. Sie möchten diese Tabelle nach unterschiedlichen Kriterien
filtern und es soll immer die Anzahl der sichtbaren Zeilen mit einer Formel gezählt
werden, die gerade trotz Filter sichtbar sind.

Tabelle mit Filterfunktion

Funktion TEILERGEBNIS() nutzen

Um diese Aufgabe zu lösen, können Sie in Excel die Funktion TEILERGEBNIS()
einsetzen. Mit der TEILERGEBNIS-Funktion können Sie ein sogenanntes
Aggregatergebnis aus einer gefilterten Liste Werte ermitteln.

TEILERGEBNIS() beherrscht viele Aggregationsfunktionen wie zum Beispiel 
SUMMME(), MITTELWERT(), ANZAHL(), MAX() etc.

Das Besondere an der TEILERGEBNIS()-Funktion ist, dass diese in gefilterten Listen nur
die sichtbaren Listeneinträge bei der Berechnung berücksichtigt. Des Weiteren kann
sie Werte in ausgeblendete Zeilen einschließen oder ausschließen.

TEILERGEBNIS() besitzt die folgende Syntax:

=TEILERGEBNIS(Funktion; Bezug1; [Bezug2]; …)

– Funktion: Eine Zahl, die angibt, welche Funktion oder Rechenoperation
verwendet werden soll; siehe folgende Tabelle

– Bezug1: Ein Bereich oder Zellbezug, auf den die Rechenoperation
angewendet werden soll

– Bezug2 (optional): Ein weiterer Bereich oder Zellbezug, auf den die
Rechenoperation angewendet wird

Da mehrere Aggregationsfunktionen (das Argument „Funktion“) für TEILERGEBNIS()
zur Verfügung stehen, sind diese in der folgenden Tabelle aufgeführt.

Mögliche Funktionen, die mit TEILERGEBNIS() eingesetzt werden können

Hinweis: „Ausgeblendet“ in den beiden Spalten dieser Übersicht bezieht sich auf die
Zeilen, die Sie aktiv ausblenden. Sie stellen dies ein über die Registerkarte Start >
Befehlsgruppe Zellen > Befehl Format > Auswahl Ausblenden und Einblenden…

Wenn Sie die Tabelle filtern, haben beide Parameter (1 oder 101 etc.) die gleiche
Wirkung.

Beispiel: Anzahl der nach Filterung angezeigten
Zeilen berechnen

Die für die Aufgabe passende Aggregationsfunktion ist die Funktion ANZAHL2(). Mit
ANZAHL2 können Sie die Anzahl der Zellen zählen, die Texte, Zahlen, Fehlerwerte und
Wahrheitswerte enthalten. Leere Zellen werden aber nicht in die Berechnung
einbezogen.

ANZAHL2() entspricht in der Funktion TEILERGEBNIS() dem Code 3 oder 103.

Die beiden Funktionscodes unterscheiden sich nur bei der Berücksichtigung von
manuell ausgeblendeten Zeilen:

– 3 = bezieht ausgeblendete Zeilen oder Zellen bei der Berechnung mit ein

– 103 = ignoriert ausgeblendete Zeilen oder Zellen bei der Berechnung

Erfassen Sie daher als Nächstes die folgende Formel in der Zelle, in der Sie die Anzahl
der gefilterten Zeilen anzeigen lassen wollen:

=TEILERGEBNIS(3; B2:B13)

– 3 = repräsentiert die Aggregationsfunktion ANZAHL2()

– B2:B13 = den Listenbereich ohne Kopfzeile, bei welchem die sichtbaren
Zellen gezählt werden sollen.

Wichtig: Da die Funktion ANZAHL2 grundsätzlich keine leeren Zellen berücksichtigt,
sollten Sie hier eine Spalte in der Liste auswählen, die keine Leerzellen enthält.

Als Ergebnis erhalten Sie den Wert 12 von der Formel geliefert. Dies ist korrekt, da alle
Datensätze im Moment in der Liste angezeigt werden.

Beispiel für die Kombination von TEILERGEBNIS() mit ANZAHL2()

Die Wirkung der Funktion TEILERGEBNIS() zeigt sich nun, wenn die Liste gefiltert wird.

Dazu wird als Filter in Spalte C eingestellt: Es sollen nur die Datensätze angezeigt
werden, deren Umsatz größer als 1.000.000 EUR ist.

Klicken Sie daher in der Spalte C = Umsatz auf das Filter-Symbol und wählen Sie die
Befehlsfolge Zahlenfilter und Größer als aus.

Tabelle filtern mit Größer als …

Es öffnet sich das Dialogfeld Benutzerdefinierter Autofilter. Erfassen Sie hier neben
ist größer als den Wert 1000000 und bestätigen Sie Ihre Einstellung, indem Sie das
Dialogfeld durch einen Klick auf OK schließen.

Einstellungen für den benutzerdefinierten Autofilter einer Excel-Tabelle

Excel filtert daraufhin die Liste und zeigt nur die Datensätze an, deren Umsatz größer
als 1.000.000 EUR ist.

Die TEILERGEBNIS-Funktion liefert jetzt den Wert 7 als Ergebnis, da nur noch die
sichtbaren Listeneinträge durch die Funktion TEILERGEBNIS() berücksichtigt werden.

Ergebnis der Funktion ANZAHL2 mit TEILERGEBNIS() und gefilterter Tabelle

Daten aus horizontaler Excel-Liste filtern

So können Sie mit der Funktion FILTER in Kombination mit MTRANS
horizontale Daten aus Zeilen filtern und in einer Spalte übersichtlich
darstellen.

Zuletzt geändert am 18.03.2026

Daten in einer horizontalen Liste – in einer Zeile

Während sich mit der Filter-Funktion von Excel vertikale Listen (von oben nach unten)
leicht auswerten lassen, ist das für horizontale Listen (von links nach rechts) nicht
ohne Weiteres mit denselben Techniken zu lösen.

Ein Beispiel: Sie haben Messwerte, Verkaufszahlen oder andere Daten, die in einer
Zeile angeordnet sind.

Nun möchten Sie aus dieser Zeile nur bestimmte Werte extrahieren – beispielsweise
alle Zahlen über einem bestimmten Schwellenwert oder nur Werte, die bestimmten
Kriterien entsprechen.

Beispiel: Daten zur Auswertung in einer horizontalen Liste

Einfache Anwendung der Funktion FILTER()

Sie möchten nur die Filialen anzeigen, deren Umsatz größer als 20.000 EUR ist. Hierfür
können Sie in Excel die FILTER-Funktion einsetzen, welche die folgende Syntax hat:

=FILTER(Matrix; Einschließen; [Wenn_leer])

Hier sind die Parameter der FILTER-Funktion erklärt. Für die Liste in der vorigen
Abbildung lautet die Funktion entsprechend:

=FILTER(B1:N2; B2:N2>20000)

Da Ihre Daten horizontal angeordnet sind, bleibt die Bedingung unverändert (>20000).
Die FILTER-Funktion gibt die gefilterten Daten jedoch ebenfalls horizontal aus.

Filter-Funktion in Excel bei horizontalen Daten

Ergebnis der Filter-Funktion in einer Spalte
darstellen

Wenn Sie die Ergebnisse vertikal darstellen möchten, können Sie dies mit der Funktion
MTRANS() lösen, indem Sie die FILTER-Funktion in diese verschachteln:

=MTRANS(FILTER(B1:N2; B2:N2>20000))

Die Funktion MTRANS() wandelt das Ergebnis von horizontal zu vertikal um, sodass die
Werte übersichtlicher dargestellt werden.

Zeilenweise Auswertung und spaltenweise Darstellung des Filterergebnisses

Jede Änderung in Ihrer Originaldatenreihe (B1:N2) wird automatisch in den gefilterten
Ergebnissen berücksichtigt.

Hinweis: Das Transponieren funktioniert auch in die andere Richtung: Die Daten aus
Ihren Spalten werden gefiltert und in (zwei) Zeilen angezeigt.

Filtern und Transponieren mit MTRANS() von Spalten nach Zeilen

Ergebnis des transponierten Filterns sortieren

Um die gefilterten Werte zusätzlich noch zu sortieren, können Sie die SORTIEREN-
Funktion einsetzen. Mit folgender Formel erhalten Sie die gefilterten Werte,
absteigend nach dem Umsatz sortiert:

=SORTIEREN(MTRANS(FILTER(B1:N2; B2:N2>20000)); 2; -1)

Sortierte und transponierte Ergebnisspalte der Filterfunktion

Mit dem Argument 2 in der Funktion SORTIEREN() legen Sie fest, dass nach der
zweiten Spalte (Umsatzwerte im Bereich A5:B6) sortiert werden soll.

Das Argument -1 definiert, dass die Werte absteigend dargestellt werden.

Hinweis: Auch das Sortieren funktioniert, wenn Sie eine vertikale Liste filtern und das
Ergebnis in Zeilen darstellen wollen. Allerdings müssen Sie hier die Sortierfunktion vor
dem Transponieren anwenden.

Sortierte und gefilterte Daten in eine Zeile transponieren

Zellen in Excel nach Farben filtern

Mit der Filterfunktion von Excel lassen sich lange Listen nicht nur nach
ihren Werten filtern, sondern auch nach ihrer Formatierung mit Zellfarbe
und Schriftfarbe. So gehen Sie vor.

Zuletzt geändert am 18.03.2026

Excel verfügt über eine mächtige Filterfunktion, mit der Sie Listen abhängig von ihren
Inhalten durchsuchen können. Sie können Listen beispielsweise nach Farben,
Hintergrundfarbe oder Schriftfarbe, filtern.

Die Wirkung dieser Funktion soll an einem Beispiel gezeigt werden.

In der folgenden Liste sind die Umsätze von verschiedenen Filialen eines Jahres
aufgeführt. Die Umsätze sind hierbei durch die Funktion Bedingte Formatierung wie
folgt eingefärbt:

– Umsätze, die über dem Durchschnittswert liegen, sind mit einem grünen
Zellhintergrund versehen.

– Umsätze, die unter dem Durchschnittswert liegen, sind mit einem roten
Zellhintergrund versehen.

Bedingte Formatierung: Zellfarbe in Abhängigkeit vom Zellwert

Wollen Sie sich jetzt alle Umsätze anzeigen lassen, die unter dem Durchschnitt liegen,
dann müssen Sie die Zellen filtern, die mit einem roten Zellhintergrund eingefärbt
sind.

Klicken Sie hierzu in der Umsatzspalte mit der linken Maustaste auf das Filtersymbol,
das sich in der Kopfzeile der Liste befindet.

Daraufhin wird ein Dialogfeld angezeigt, in dem Sie Einstellungen zum Filtern und
Sortieren der Liste vornehmen können. Wählen Sie hier den Eintrag Nach Farbe
filtern > Nach Zellenfarbe filtern > Symbol für rote Farbe aus.

Listen in Excel nach Farben filtern

Aufgrund dieser Auswahl wird die Liste automatisch nach den Zellen mit der roten
Hintergrundfarbe gefiltert.

Hinweis: Wie Sie in der vorherigen Abbildung sehen, können Sie die Zellen nicht nur
nach der Hintergrundfarbe, sondern auch nach der Schriftfarbe filtern. Verwenden Sie
hierzu statt Nach Zellenfarbe filtern den Eintrag Nach Schriftfarbe filtern und
wählen hier die entsprechende Schriftfarbe aus.

Ergebnis: Listenauszug mit rot hinterlegten Zellwerten (Auszug)

Die Filterung funktioniert nicht nur bei der Bedingten Formatierung der Zellen. Mit
dieser Filterfunktion können Sie alle Listen und Tabellen filtern, die Sie mit
unterschiedlichen Farben (manuell) eingefärbt haben.

